


Analysis of Average Daily Gain Data

An animal scientist compared two diets to see if one is better than the other in terms of enhancing weight gain of cattle. The cattle were divided into two groups of 16 animals (bulls and steers). Half of each group were given diet 1 and the other half were given diet 2. The average daily weight gain (adg) was measured for each animal.

From the interaction plot of mean (below), it is evident that steers gain more weight than bulls. For the bulls, diet 1 has a somewhat larger mean but the opposite is true for steers. Thus there are questions whether or not there is interaction and whether or not there is a difference between diet means for bulls and for steers.



Even though the lines in the interaction plot are not parallel, results from the interaction model indicate that the interaction is not significant ($F=0.53$, $p=0.4739$). Therefore, we can examine the main effects for Group and Diet. There is not a significant difference between the diets ($F=0.15$, $p=0.7007$), but the difference between bulls and steers is significant ($F=16.45$, $p=0.0004$). The steers gain, on average, 0.80 pounds per day more than bulls.

SAS Code and Annotated Output

```
data DietData;
input group $ diet adg @@;
datalines;
Bull 1 1.87    Bull 1 1.60    Bull 1 1.76    Bull 1 2.13
Bull 1 1.81    Bull 1 2.64    Bull 1 1.33    Bull 1 1.80
Steer 1 1.12    Steer 1 3.20    Steer 1 2.72    Steer 1 3.95
Steer 1 2.03    Steer 1 2.29    Steer 1 2.83    Steer 1 2.03
Bull 2 1.49    Bull 2 2.08    Bull 2 2.29    Bull 2 1.55
Bull 2 1.39    Bull 2 1.71    Bull 2 1.98    Bull 2 1.92
Steer 2 2.61    Steer 2 2.67    Steer 2 3.20    Steer 2 2.72
Steer 2 2.40    Steer 2 1.87    Steer 2 2.93    Steer 2 3.52
;
run;

proc sort data=DietData;
by group diet;
run;

proc means data=DietData noprint;
by group diet;
var adg;
output out=means mean=mean;
run;

proc sgplot data=means;
series x=diet y=mean / group=group;
run;

proc glm data=DietData;
class group diet;
model adg = group diet group*diet / ss3;
lsmeans group diet group*diet / stderr pdiff;
run;

quit;
```

The first part of the GLM output gives the levels of each factor and the total number of observations used to fit the model. You should always look at this information and make sure it is correct.

| The GLM Procedure | | |
|-----------------------------|--------|------------|
| Class Level Information | | |
| Class | Levels | Values |
| group | 2 | Bull Steer |
| diet | 2 | 1 2 |
| Number of Observations Read | | 32 |
| Number of Observations Used | | 32 |

The ANOVA table shows

- (1) The group*diet interaction is not significant, so we can examine the main effects.
- (2) The Diet main effect is not significant, so we do not look at any detail regarding the differences in average daily weight gain for the two diets.
- (3) The Group main effect is significant, so we take a closer look at the marginal means for the two groups.

| Dependent Variable: adg | | | | | |
|-------------------------|-----------|----------------|-------------|---------|--------|
| Source | DF | Sum of Squares | Mean Square | F Value | Pr > F |
| Model | 3 | 5.28107500 | 1.76035833 | 5.71 | 0.0035 |
| Error | 28 | 8.63292500 | 0.30831875 | | |
| Corrected Total | 31 | 13.91400000 | | | |
| R-Square | Coeff Var | Root MSE | adg Mean | | |
| 0.379551 | 24.87187 | 0.555265 | 2.232500 | | |
| Source | DF | Type III SS | Mean Square | F Value | Pr > F |
| group | 1 | 5.07211250 | 5.07211250 | 16.45 | 0.0004 |
| diet | 1 | 0.04651250 | 0.04651250 | 0.15 | 0.7007 |
| group*diet | 1 | 0.16245000 | 0.16245000 | 0.53 | 0.4739 |

The means that need to be interpreted are the marginal means of Group. We see a p-value of 0.0004 for the difference of means. The estimated difference of means is $2.63 - 1.83 = 0.80$ pounds. The steers gained an average of 0.80 pounds per day more than bulls.

| group | adg LSMEAN | Standard Error | H0:LSMEAN=0 Pr > t | H0:LSMean1=LSMean2 Pr > t |
|-------|------------|----------------|------------------------|-------------------------------|
| Bull | 1.83437500 | 0.13881614 | <.0001 | 0.0004 |
| Steer | 2.63062500 | 0.13881614 | <.0001 | |

We do not report any of the following information, since the main effect for Diet was not significant.

| diet | adg LSMEAN | Standard Error | H0:LSMEAN=0 Pr > t | H0:LSMean1=LSMean2 Pr > t |
|------|------------|----------------|------------------------|-------------------------------|
| 1 | 2.19437500 | 0.13881614 | <.0001 | 0.7007 |
| 2 | 2.27062500 | 0.13881614 | <.0001 | |

We do not report any of the following information, since the interaction was not significant.

| group | diet | adg LSMEAN | Standard Error | Pr > t | LSMEAN Number |
|-------|------|------------|----------------|---------|---------------|
| Bull | 1 | 1.86750000 | 0.19631567 | <.0001 | 1 |
| Bull | 2 | 1.80125000 | 0.19631567 | <.0001 | 2 |
| Steer | 1 | 2.52125000 | 0.19631567 | <.0001 | 3 |
| Steer | 2 | 2.74000000 | 0.19631567 | <.0001 | 4 |

Least Squares Means for effect group*diet
Pr > |t| for H0: LSMean(i)=LSMean(j)

Dependent Variable: adg

| i/j | 1 | 2 | 3 | 4 |
|-----|--------|--------|--------|--------|
| 1 | | 0.8131 | 0.0258 | 0.0039 |
| 2 | 0.8131 | | 0.0149 | 0.0021 |
| 3 | 0.0258 | 0.0149 | | 0.4374 |
| 4 | 0.0039 | 0.0021 | 0.4374 | |

NOTE: To ensure overall protection level, only probabilities associated with pre-planned comparisons should be used.