Annotated MIXED code and output

Data step and code for MIXED procedure

The data are in the text file "hospital data". Here Patient denotes the patient number, hosp denotes the hospital number, pro denotes the medical procedure (a or b), and recover denotes the patient recovery time.

```
data;
input Patient hosp pro $ recover;
datalines;
1     1     a     10
2     1     a     12
3     1     a     18
   etc.
22    4     b     24
23    4     b     28
24    4     b     30
;
proc mixed;
class hosp pro;
model recover = pro;
random hosp pro*hosp;
lsmeans pro/pdiff;
run;
```

Differences between MIXED and GLM two-way ANOVA code

MIXED has a model statement for fixed effects and a random statement for random effects.

```
model recover = pro;
random hosp pro*hosp;
```

GLM has just a model statement in which all terms on the right side of the equal sign are assumed to be fixed effects.

```
model recover = pro hosp pro*hosp;
```

The MIXED pdiff option computes differences of means, standard errors of differences, t-statistics for differences of means, and p-values for the t-statistics.

```
lsmeans pro/pdiff;
```

The GLM pdiff stderr options compute the standard errors and a table of pairwise p-values.

```
lsmeans pro/stderr pdiff;
```

Output

The MIXED output is larger than the GLM output.

Here is up-front information that is not essential in interpreting the output. It identifies the dependent variable (we call it the response variable) which is “recover” in this case, and it has information about the computational procedure used by MIXED.

```
The Mixed Procedure

                  Model Information

Data Set                      WORK.DATA2
Dependent Variable            recover
Covariance Structure          Variance Components
Estimation Method             REML
Residual Variance Method      Profile
Fixed Effects SE Method       Model-Based
Degrees of Freedom Method     Containment

             Class Level Information

Class    Levels    Values

hosp          4    1 2 3 4
pro           2    a b

            Dimensions

Covariance Parameters             3
Columns in X                      3
Columns in Z                     12
Subjects                          1
Max Obs Per Subject              24

          Number of Observations

Number of Observations Read              24
Number of Observations Used              24
Number of Observations Not Used           0

                    Iteration History

Iteration    Evaluations    -2 Res Log Like       Criterion

        0              1       133.20480472
        1              1       127.45438333      0.00000000

                  Convergence criteria met.
```

These are the estimates of the variances of the random effects. The “residual” is the same as our “error”.

Covariance Parameter Estimates	
Cov Parm	Estimate
hosp	2.9074
hosp*pro	8.8935
Residual	10.2500

These are measures of how well the model fits the data. We will not consider them.

Fit Statistics	
-2 Res Log Likelihood	127.5
AIC (smaller is better)	133.5
AICC (smaller is better)	134.8
BIC (smaller is better)	131.6

The next three parts of the output give the information that we need to determine statistical significance. These are the most important parts for us.

Here is the test of significance of the fixed effects. It shows that there are significant differences among the means of “pro” (p = .0281).

Type 3 Tests of Fixed Effects

Effect	Num DF	Den DF	F Value	Pr > F
pro	1	3	15.98	0.0281

Here are the lsmeans and the standard errors of the fixed effects. The p-values test whether or not the population means of the fixed effects are 0. Such tests are usually not of interest.

Least Squares Means

Effect	pro	Estimate	Standard Error	DF	t Value	Pr > |t|
pro	a	12.2500	1.9505	3	6.28	0.0082
pro	b	22.1667	1.9505	3	11.36	0.0015

Here are comparisons of difference of means. “Estimate” is the difference of the means, “Standard Error” is the standard error of the difference. "DF" is the degrees of freedom of the t-statistic, t-Value is the t-statistic for testing the difference of the means, and "Pr > |t|" is the p-value for the t-statistic. It tells us in this case that the means are significantly different at the 5% level (p = .0281). If there are more than two levels, all differences of means of the fixed effects will be displayed in this table.

Differences of Least Squares Means

Effect	pro	_pro	Estimate	Standard Error	DF	t Value	Pr > |t|
pro	a	b	-9.9167	2.4809	3	-4.00	0.0281